
*    VIN      R0 100K
*     ______/\  /\  /\__      IABC = 1mA -> 0
*    |        \/  \/    |       ^
*    |                  | +    /_\
*   _|_     _/\  /\  /\_|___|\  |   OUT  |\ VOUT
*  /_  \   _|_ \/  \/   INP | \/\/\______| \____
* // \  \  \\\        ______| /\/\/ _|_  | /    |
* \   \//     R1 200  |   _ |/      ___  |/     |
*  \___/           INN|    B_OTA1    |C1        |
*   _|_               |             _|_  B_BUF1 |
*   ///       R2      |       R3    \\\         |
*         _/\  /\  /\_|_/\  /\  /\______________|
*       _|_  \/  \/       \/  \/
*       ///   200        100K
*

VIN       VIN        0      PULSE( -1  1 1u 1u 1u  .2m .4m )  AC 1
R0        VIN        INP   100K
R1        INP        0     200
R2        INN        0     200
R3        VOUT       INN   100K
B_OTA1    OUT        0     I =   -1*v(VIABC)*tanh((v(INP)-v(INN))/.052)
C1        OUT        0     150p
Rneed     OUT        0     10000k
B_BUF1    VOUT       0     V =    v(OUT) -1.2
V_Iabc    VIABC      0     PWL    ( 0 1m 2m 0 ) 

.control
tran      1u         2m   0      1u   
set       pensize =  2
plot       v(vout)   v(vin)    
ac        dec        10    1       1000K 
plot      db(out)     title IABC1000uA
alter     V_Iabc     dc =  10u
ac        dec        10    1       1000K  
plot      db(out)    title IABC10uA
.endc

.end

=====================END_OF_SPICE============================

A voltage control Low pass filter is a the typical application 
given in the data sheet. At one time this actual filter was
being used as an automotive alternative to the dolby noise 
reduction circuitry applied to cassette tapes. A home project
using the LM13700 actually led to the development of "DNR".

http://en.wikipedia.org/wiki/Dynamic_Noise_Reduction

The concept is when music has a high frequency content,
set the bandwidth to be high..

But when the bandwidth of the music content is low,
the bandwidth can be reduced as well. The noise
of the music off the cassette will appear to be much lower.



Since the LM13700 is stereo, this was a natural for 
a home HIFI project. The total circuit only needed
one LM13700 and one LM324. 
 

But there may be another application for this voltage controlled
low pass filter. This circuit is also a voltage follower with
adjustable settling time. The waveform above shows how the 
low pass follows a square wave as the IABC level moves from 1mA
to zero over 2msec. The time constant is tracking the inverse
of IABC such that it is longest when IABC is lowest. This may be 
of use for voltage control amplifier applications where 
flexibility  in the attack and decay of the envelope is desired. 

